

NAG Library Routine Document

G01GBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G01GBF returns the lower tail probability for the noncentral Student's t -distribution, via the routine name.

2 Specification

```
FUNCTION G01GBF (T, DF, DELTA, TOL, MAXIT, IFAIL)
REAL (KIND=nag_wp) G01GBF
INTEGER MAXIT, IFAIL
REAL (KIND=nag_wp) T, DF, DELTA, TOL
```

3 Description

The lower tail probability of the noncentral Student's t -distribution with ν degrees of freedom and noncentrality parameter δ , $P(T \leq t : \nu; \delta)$, is defined by

$$P(T \leq t : \nu; \delta) = C_\nu \int_0^\infty \left(\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\alpha u - \delta} e^{-x^2/2} dx \right) u^{\nu-1} e^{-u^2/2} du, \quad \nu > 0.0$$

with

$$C_\nu = \frac{1}{\Gamma\left(\frac{1}{2}\nu\right) 2^{(\nu-2)/2}}, \quad \alpha = \frac{t}{\sqrt{\nu}}.$$

The probability is computed in one of two ways.

(i) When $t = 0.0$, the relationship to the normal is used:

$$P(T \leq t : \nu; \delta) = \frac{1}{\sqrt{2\pi}} \int_\delta^\infty e^{-u^2/2} du.$$

(ii) Otherwise the series expansion described in Equation 9 of Amos (1964) is used. This involves the sums of confluent hypergeometric functions, the terms of which are computed using recurrence relationships.

4 References

Amos D E (1964) Representations of the central and non-central t -distributions *Biometrika* **51** 451–458

5 Parameters

- 1: T – REAL (KIND=nag_wp) *Input*
On entry: t , the deviate from the Student's t -distribution with ν degrees of freedom.
- 2: DF – REAL (KIND=nag_wp) *Input*
On entry: ν , the degrees of freedom of the Student's t -distribution.
Constraint: DF \geq 1.0.

- 3: DELTA – REAL (KIND=nag_wp) Input
On entry: δ , the noncentrality parameter of the Students t -distribution.
- 4: TOL – REAL (KIND=nag_wp) Input
On entry: the absolute accuracy required by you in the results. If G01GBF is entered with TOL greater than or equal to 1.0 or less than $10 \times$ *machine precision* (see X02AJF), then the value of $10 \times$ *machine precision* is used instead.
- 5: MAXIT – INTEGER Input
On entry: the maximum number of terms that are used in each of the summations.
Suggested value: 100. See Section 8 for further comments.
Constraint: $\text{MAXIT} \geq 1$.
- 6: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on exit IFAIL \neq 0, then G01GBF returns 0.0.

IFAIL = 1

On entry, $DF < 1.0$.

IFAIL = 2

On entry, $\text{MAXIT} < 1$.

IFAIL = 3

One of the series has failed to converge. Reconsider the requested tolerance and/or maximum number of iterations.

IFAIL = 4

The probability is too small to calculate accurately.

7 Accuracy

The series described in Amos (1964) are summed until an estimated upper bound on the contribution of future terms to the probability is less than TOL. There may also be some loss of accuracy due to calculation of gamma functions.

8 Further Comments

The rate of convergence of the series depends, in part, on the quantity $t^2/(t^2 + \nu)$. The smaller this quantity the faster the convergence. Thus for large t and small ν the convergence may be slow. If ν is an integer then one of the series to be summed is of finite length.

If two tail probabilities are required then the relationship of the t -distribution to the F -distribution can be used:

$$F = T^2, \lambda = \delta^2, \nu_1 = 1 \quad \text{and} \quad \nu_2 = \nu,$$

and a call made to G01GDF.

Note that G01GBF only allows degrees of freedom greater than or equal to 1 although values between 0 and 1 are theoretically possible.

9 Example

This example reads values from, and degrees of freedom for, and noncentrality parameters of the noncentral Student's t -distributions, calculates the lower tail probabilities and prints all these values until the end of data is reached.

9.1 Program Text

```

Program g01gbfe

!      G01GBF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g01gbf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: delta, df, prob, t, tol
Integer                     :: ifail, maxit
!      .. Executable Statements ..
Write (nout,*) 'G01GBF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Display titles
Write (nout,*) '      T      DF      DELTA      PROB'
Write (nout,*)

!      Use default tolerance and suggested number of iterations
tol = 0.0E0_nag_wp
maxit = 100

d_lp: Do
  Read (nin,*,Iostat=ifail) t, df, delta
  If (ifail/=0) Then
    Exit d_lp
  End If

!      Calculate probability
  ifail = 0
  prob = g01gbf(t,df,delta,tol,maxit,ifail)

!      Display results

```

```
      Write (nout,99999) t, df, delta, prob
      End Do d_lp

99999 Format (1X,3F8.3,F8.4)
      End Program g01gbfe
```

9.2 Program Data

```
G01GBF Example Program Data
-1.528  20.0  2.0           :T DF DELTA
-0.188   7.5  1.0           :T DF DELTA
 1.138  45.0  0.0           :T DF DELTA
```

9.3 Program Results

```
G01GBF Example Program Results
```

T	DF	DELTA	PROB
-1.528	20.000	2.000	0.0003
-0.188	7.500	1.000	0.1189
1.138	45.000	0.000	0.8694
